

Computer aids short-cut distillation design

A computer program is offered to handle separation of key components in a distillation design by applying the method of Newton-Raphson

Huan-Yang Chang, University of Lowell, Lowell, Mass.

IF PRODUCT PURITY for multicomponent distillation were specified in terms of the mole fraction of heavy key component in the light product (and light key component in the heavy product), a trial-and-error calculation would be required. A computer program is presented for making the calculations using the Newton-Raphson iterative method.

Yaws et al¹ derived short-cut equations to handle specifications for light key component in the light fraction (and heavy key component in the heavy fraction). But for the complementary cases to be discussed here, recoveries of the unspecified key components will be given assumed values in order to carry forth the calculations.

Method derived. Assume that the feed composition and the average values of the relative volatilities are known. We start with a set of guessed values of the recoveries of the key component in the distillate and bottoms. The number of moles of the keys are established by the following relations:

$$d_{LK} = (d_{LK}/f_{LK})(f_{LK}) \quad (1)$$

$$b_{LK} = f_{LK} - d_{LK} \quad (2)$$

$$d_{HK} = [1 - (b_{HK}/f_{HK})](f_{HK}) \quad (3)$$

$$b_{HK} = f_{HK} - d_{HK} \quad (4)$$

The distributions of the non-keys are estimated by the Hengstebeck-Geddes^{2,3} equation.

$$\ln(d_i/b_i) = C_1 + C_2 \ln \alpha_i \quad (5)$$

Substituting the *i*-th component in Eq. 5 for *LK* and *HK*, we obtain two equations in two unknowns, *C*₁ and *C*₂, which can be solved simultaneously.

$$C_1 = \ln(d_{HK}/b_{HK}) \quad (6)$$

$$C_2 = \ln[(d_{LK}/b_{LK})(b_{HK}/d_{HK})]/\ln \alpha_{LK} \quad (7)$$

Eq. 5 can be written as

$$d_i/b_i = \text{Exp}(C_1 + C_2 \ln \alpha_i) \quad (5')$$

The material balance of component *i* is

$$f_i = d_i + b_i \quad (8)$$

Eliminating *d_i* between 5' and 8, we have

$$b_i = \frac{f_i}{1 + \text{Exp}(C_1 + C_2 \ln \alpha_i)} \quad (9)$$

$$d_i = f_i - b_i \quad (10)$$

The mole fractions of each component in the distillate and the bottoms are computed as

$$x_{Di} = d_i / \sum d_i \quad (11)$$

$$\text{and } x_{Bi} = b_i / \sum b_i \quad (12)$$

Let *s*₁ = *d*_{LK}/*f*_{LK} and *s*₂ = *b*_{HK}/*f*_{HK}.

We define two functions g_1 and g_2 by

$$g_1(s_1, s_2) = (x_{D,HK})_{ca} - x_{D,HK} \quad (13)$$

$$g_2(s_1, s_2) = (x_{B,LK})_{ca} - x_{B,LK} \quad (14)$$

where $(x_{D,HK})_{ca}$ and $(x_{B,LK})_{ca}$ are the mole fractions of the key components calculated by Eqs. 11 and 12, based on assumed values of s_1 and s_2 , while $x_{D,HK}$ and $x_{B,LK}$ are the specified mole fractions.

The objective now is to find values of s_1 and s_2 so that both functions g_1 and g_2 will become zero. For this purpose, the Newton-Raphson method is useful. Suppose $s_{1,old}$ and $s_{2,old}$ are reasonable approximations and it is wished to obtain new (hopefully better) approximations $s_{1,new}$ and $s_{2,new}$. Newton-Raphson method imposes the following procedure:

$$s_{1,new} = s_{1,old} + \Delta s_1 \quad (15)$$

$$s_{2,new} = s_{2,old} + \Delta s_2 \quad (16)$$

Δs_1 and Δs_2 can be evaluated by solving the following

$$\begin{bmatrix} g'_{1,1} & g'_{1,2} \\ g'_{2,1} & g'_{2,2} \end{bmatrix} \begin{bmatrix} \Delta s_1 \\ \Delta s_2 \end{bmatrix} = \begin{bmatrix} -g_1 \\ -g_2 \end{bmatrix} \quad (17)$$

where $g'_{i,j} = \partial g_i / \partial s_j$ (18)

The implicit nature of the g functions implies that the partial derivatives cannot be obtained analytically. It is more convenient to employ a finite-difference approximation method.

$$\frac{\partial g_i}{\partial s_1} = \frac{g_i(s_1 + \delta, s_2) - g_i(s_1, s_2)}{\delta}$$

and

$$\frac{\partial g_i}{\partial s_2} = \frac{g_i(s_1, s_2 + \delta) - g_i(s_1, s_2)}{\delta} \quad (19)$$

for

$$i = 1 \text{ and } 2.$$

We have $\delta = 10^{-5}$ as a satisfactory value. The updating of s_1 and s_2 is repeated using Eqs. 15, 16, and 17 until some desired degree of convergence is obtained. The criteria are arbitrarily chosen for the differences between the expected and calculated mole fractions of both key components to be within ± 0.0001 .

Computer program. A listing of the computer PROGRAM HGFUG is presented in the accompanying box. The main program solves the distribution of the keys in the distillate and the bottoms by the Newton-Raphson iteration technique. It then calls SUBROUTINE HG to establish the distribution of the non-keys



About the author

HUAN-YANG CHANG is associate professor in the Dept. of Chemical Engineering for University of Lowell, Lowell, Mass. Dr. Chang received his Ph.D. degree in chemical engineering from Iowa State University. He has been a chemical engineer for Taiwan Paper Industry, Inc. and a research chemical engineer for Halby Chemical Co. He is a member of AIChE.

CASE A

RHE

T 2, 3, 4, 1, 1.5, .10, .15
T .25, .25, .25, .25
T 3.86, 1.89, 1, .65
T .9, .9

*** INPUT ***

TOTAL NO. OF COMPONENTS = 4
LIGHT KEY IS COMPONENT NO. 2
HEAVY KEY IS COMPONENT NO. 3
FEED THERMAL CONDITION $\theta = 1.0$
THE OPERATING REFLUX RATIO IS 1.50 TIMES RMIN

THE FEED COMPOSITION AND THE RELATIVE VOLATILITIES ARE

I	XF(I)	ALFA(I)
1	.2500	3.8600
2	.2500	1.8900
3	.2500	1.0000
4	.2500	.6500

THE SPECIFIED MOLE FRACTIONS OF THE KEYS IN THE DISTILLATE AND IN THE BOTTOMS ARE

XD(HK) = .1000 XB(LK) = .1500

THE INITIALLY GUESSED VALUES OF THE DISTRIBUTIONS ARE

DFLK = .9000 BFNK = .9000

*** OUTPUT ***

AFTER 3 ITERATIONS THE ACCEPTED VALUES OF THE DISTRIBUTIONS ARE

DFLK = .6825 BFNK = .8118

I	D(I)	XB(I)	B(I)	XB(I)
1	.24079E+02	.51136E+00	.92084E+00	.17377E-01
2	.17064E+02	.34232E+00	.79343E+01	.14994E+00
3	.47050E+01	.99989E-01	.20294E+02	.38342E+00
4	.12214E+01	.25950E-01	.23779E+02	.44924E+00
TOTAL	.47070E+02		.52930E+02	

MINIMUM NUMBER OF PLATES = 3.50

PHI = 1.280435 RMIN = .504

AT R = .759 NO. OF T. P. = 8.194

SRU 2.375 UNITS.

RUN COMPLETE.

by the Hengstebeck-Geddes^{2,3} equation. The minimum number of theoretical plates is based on Fenske's equation.⁴

$$N_{min} = \frac{\ln[(x_{D,LK}/x_{B,LK})(x_{B,HK}/x_{D,HK})]}{\ln \alpha_{LK}} \quad (20)$$

Since mole ratio equals mole fraction ratio, N_{min} in Fenske's equation is equivalent to C_2 in Eq. 7. SUBROUTINE UWD then calculates the minimum reflux ratio by the Underwood's equations.⁵

$$1 - q = \sum_i^n \frac{\alpha_i x_{Fi}}{\alpha_i - \phi} \quad (21)$$

Eq. 21 is solved for $\alpha_{HK} < \phi < \alpha_{LK}$ by the bisection method, which always converges to the desired solution. The value of ϕ is substituted into Eq. 22 to obtain R_{min} .

$$R_{min} + 1 = \sum_i^n \frac{\alpha_i x_{Di}}{\alpha_i - \phi} \quad (22)$$

Finally, SUBROUTINE GLLD calculates the number of theoretical plates at operating reflux ratio by the Gilliland's correlation.⁶ Molokanov, Korabline, Mazurina, and Nikiforov⁷ have fitted Gilliland's correlation curve into the following equation:

$$Y = 1 - \text{Exp} \left[\frac{(1 + 54.4X)(X - 1)}{(11 + 117.2X) \sqrt{X}} \right] \quad (23)$$

where

$$X = \frac{R - R_{min}}{R + 1} \text{ and } Y = \frac{N - N_{min}}{N + 1}$$

CASE B

```

RHH
* 2, 3, 4, 1., 1.5, .05, .05
* .25, .25, .25, .25
* 3.86, 1.89, 1., .65
* .9, .9

*** INPUT ***

TOTAL NO. OF COMPONENTS = 4
LIGHT KEY IS COMPONENT NO. 2
HEAVY KEY IS COMPONENT NO. 3
FEED THERMAL CONDITION Q = 1.0
THE OPERATING REFLUX RATIO IS 1.50 TIMES RMIN

THE FEED COMPOSITION AND THE RELATIVE VOLATILITIES ARE
I  X(FI)  ALFA(I)
1  .2500  3.8600
2  .2500  1.8900
3  .2500  1.0000
4  .2500  .6500

THE SPECIFIED MOLE FRACTIONS OF THE KEYS IN THE DISTILLATE
AND IN THE BOTTOMS ARE
XD(HK) = .0500  XB(LK) = .0500

THE INITIALLY GUESSED VALUES OF THE DISTRIBUTIONS ARE
DFLK = .9000  BFHK = .9000

*** OUTPUT ***

AFTER 2 ITERATIONS THE ACCEPTED VALUES OF THE DISTRIBUTIONS ARE
DFLK = .9003  BFHK = .8997

I  D(I)  XD(I)  B(I)  XB(I)
1  .24990E+02  .4982E+00  .20006E-01  .40121E-03
2  .22527E+02  .44892E+00  .24932E+01  .50000E-01
3  .25040E+01  .50000E-01  .22493E+02  .45100E+00
4  .14159E+00  .28241E-02  .24858E+02  .49852E+00
TOTAL  .50132E+02  .49846E+02

MINIMUM NUMBER OF PLATES = 6.90
PHI = 1.280435  RMIN = .956
AT R = 1.434  NO. OF T. P. = 12.733

SRU 2.366 UNITS.
RUN COMPLETE.

```

CASE C

```

RHH
* 2, 3, 4, 1., 1.5, .0005, .0005
* .25, .25, .25, .25
* 3.86, 1.89, 1., .65
* .9, .9

*** INPUT ***

TOTAL NO. OF COMPONENTS = 4
LIGHT KEY IS COMPONENT NO. 2
HEAVY KEY IS COMPONENT NO. 3
FEED THERMAL CONDITION Q = 1.0
THE OPERATING REFLUX RATIO IS 1.50 TIMES RMIN

THE FEED COMPOSITION AND THE RELATIVE VOLATILITIES ARE
I  X(FI)  ALFA(I)
1  .2500  3.8600
2  .2500  1.8900
3  .2500  1.0000
4  .2500  .6500

THE SPECIFIED MOLE FRACTIONS OF THE KEYS IN THE DISTILLATE
AND IN THE BOTTOMS ARE
XD(HK) = .0005  XB(LK) = .0005

THE INITIALLY GUESSED VALUES OF THE DISTRIBUTIONS ARE
DFLK = .9000  BFHK = .9000

*** OUTPUT ***

AFTER 3 ITERATIONS THE ACCEPTED VALUES OF THE DISTRIBUTIONS ARE
DFLK = .9990  BFHK = .9990

I  D(I)  XD(I)  B(I)  XB(I)
1  .25000E+02  .50000E+00  .46574E-08  .93152E-10
2  .24972E+02  .49950E+00  .24990E-01  .49979E-03
3  .24989E-01  .49979E-03  .24975E+02  .49950E+00
4  .21788E-05  .43577E-07  .25000E+02  .50000E+00
TOTAL  .50000E+02  .50000E+02

MINIMUM NUMBER OF PLATES = 21.70
PHI = 1.200435  RMIN = 1.295
AT R = 1.943  NO. OF T. P. = 39.796

SRU 2.357 UNITS.
RUN COMPLETE.

```

Example Problems. In a multicomponent distillation the feed is a saturated liquid ($q = 1$) having the following composition and volatility.

Component	Mole fraction	Relative volatility
1	.25	3.86
2 (LK)	.25	1.89
3 (HK)	.25	1.00
4	.25	0.65

Case a. Obtain a solution for the specification

$$\begin{aligned}
 x_{D,HK} &= .10 \\
 x_{B,LK} &= .15 \\
 R/R_{min} &= 1.5
 \end{aligned}$$

Case b. Repeat for the specification

$$\begin{aligned}
 x_{D,HK} &= .05 \\
 x_{B,LK} &= .05 \\
 R/R_{min} &= 1.5
 \end{aligned}$$

Case c. Repeat for the specification

$$\begin{aligned}
 x_{D,HK} &= .0005 \\
 x_{B,LK} &= .0005 \\
 R/R_{min} &= 1.5
 \end{aligned}$$

Case a represents relatively low recoveries while Case c represents very high recoveries. It can be seen from the computer printout that all three cases have converged rapidly to the desired solutions. The same initial guess has been used for all cases, i.e., 90 percent recovery of the light key in the distillate and 90 percent recovery of the heavy key in the bottoms.

NOMENCLATURE

b	number of moles of a component in the bottoms
C_1, C_2	constants in Hengstebeck-Geddes equation
d	number of moles of a component in the distillate
f	number of moles of a component in the feed
g	functions defined by Eqs. 13 and 14
g'	partial derivatives of g defined by Eq. 18
N	number of theoretical plates
N_{min}	minimum number of theoretical plates at total reflux
q	pseudo-ratio of liquid to vapor for the feed
R	reflux ratio
R_{min}	minimum reflux ratio
x_1	fraction recovery of the light key in the distillate
x_2	fraction recovery of the heavy key in the bottoms
x	mole fraction of a component where x_B, x_D, x_F = mole fractions in the bottoms, distillate, and feed, respectively
α	relative volatility with respect to the heavy key
δ	arbitrarily assigned small positive difference
ϕ	parameter in Underwood's equations
Σ	summation

SUBSCRIPTS

HK, LK heavy and the light key component individual component identifier

LITERATURE CITED

- 1 Yaws, C. L., Patel, P. M., Pitts, F. H., and Fang, C. S., *Hydrocarbon Processing*, Vol. 58, No. 2, Feb. 1979, p. 99.
- 2 Hengstebeck, R. J., *Trans. A.I.Ch.E.*, Vol. 42, 1946, p. 309.
- 3 Geddes, R. L., *A.I.Ch.E.J.*, Vol. 4, 1958, p. 389.
- 4 Fenske, M. R., *Ind. Eng. Chem.*, Vol. 24, 1932, p. 482.
- 5 Underwood, A. J. V., *Chem. Eng. Prog.*, Vol. 44, 1948, p. 605.
- 6 Gilliland, E. R., *Ind. Eng. Chem.*, Vol. 32, 1940, p. 1101.
- 7 Mukhanov, Y. K., Korzhine, T. P., Mazurina, N. I., and Nikiforov, G. A., *International Chemical Engineering*, Vol. 12, No. 2, 1972, p. 209.

Programa en FORTRAN

```
PROGRAM HGFUGK
C
C HENGSTEBECK-GEDDES-FENSKE-UNDERWOOD-GILLILAND-KIRKBRIDE
C SHORT-CUT DISTILLATION METHOD
C ALL INPUT ENTRIES ARE FORMAT FREE.
C THE INPUT CONSIST OF THE FOLLOWING:
C 1.- LK,HK,NOC,Q,RRM,XDHK,XBLK
C 2.- XF(I)
C 3.- ALFA(I)
C 4.- DFLK,BFHK
C DEFINITIONS:
C LK=NO. ASSIGNED TO THE LIGHT KEY COMPONENT
C HK=NO. ASSIGNED TO THE HEAVY KEY COMPONENT
C NOC= TOTAL NO. OF COMPONENTS
C Q=FEED THERMAL CONDITION
C RRM=R/RMIN, WHERE R = OPERATING REFLUX RATIO AND
C RMIN=MINIMUM REFLUX RATIO
C XDHK=MOLE FRACTION OF THE HEAVY KEY IN THE DISTILLATE
C XBLK=MOLE FRACTION OF THE LIGHT KEY IN THE BOTTOMS
C XF(I)=MOLE FRACTION OF A COMPONENT IN THE FEED
C ALFA(I)=AVERAGE VALUE OF THE RELATIVE VOLATILITY OF A COMPONENT.
C DFLK=INITIAL GUESS OF FRACTION RECOVERY OF THE LIGHT
C KEY COMPONENT IN THE DISTILLATE
C BFHK=INITIAL GUESS OF FRACTION RECOVERY OF THE HEAVY
C KEY COMPONENT IN THE BOTTOMS
C FLOW RATES ARE BASED ON F=100 MOLES
C THE OUTPUT PRINTS OUT THE FOLLOWING:
C 1. NO. OF ITERATIONS IN THE NEWTON-RAPHSON METHOD
C 2. CONVERGED VALUES OF DFLK AND BFHK
C 3. DISTILLATE AND BOTTOMS COMPOSITIONS
C 4. NMIN BY THE FENSKE'S EQUATION
C 5. PHI AND RMIN BY THE UNDERWOOD'S EQUATIONS
C 6. NO. OF T.P. AT SPECIFIED R BY GILLILAND'S CORRELATIONS
C 7. FEED PLATE LOCATION BY KIRKBRIDE'S CORRELATION
C
REAL N,NMIN,NF
INTEGER LK, HK
DIMENSION FUNC(2),S(2),G(2),DG(2,2)
COMMON/BLOCK1/XF(15),XD(15),ALFA(15)
COMMON/BLOCK2/FD,SUMB,SUMD,D(15),B(15),F(15),XB(15)
COMMON NOC,LK,HK
EQUIVALENCE (C2,NMIN),(S(1),DFLK),(S(2),BFHK)
C
C** DATA INPUT AND PRINTOUT
C
READ*,LK,HK,NOC,Q,RRM,XDHK,XBLK
READ*,(XF(I),I=1,NOC)
READ*,(ALFA(I),I=1,NOC)
READ*,DFLK,BFHK
WRITE(6,900)
WRITE(6,901)NOC
WRITE(6,902)LK,HK
WRITE(6,903)Q
WRITE(6,920)RRM
WRITE(6,904)
WRITE(6,905)(I,XF(I),ALFA(I),I=1,NOC)
WRITE(6,906)XDHK,XBLK
C
C** CALCULATION BASIS
C
FD=100.
C
C** CARRY OUT NEWTON-RAPHSON ITERATION.
```

```

C
  KMAX=20
  DELTA=1.E-5
  DO 100 K=1,KMAX
  CALL HG(S(1),S(2),C2)
  FUNC(1)=XD(HK)-XDHK
  FUNC(2)=XB(LK)-XBLK
  IF (ABS(FUNC(1)).LT..0001.AND.ABS(FUNC(2)).LT..0001) GO TO 200
C
C** ESTIMATE THE FOUR PARTIAL DERIVATIVES.
C
  DO 5 I=1,2
  5 G(I)=FUNC(I)
  DO 20 I=1,2
  S(I)=S(I)+DELTA
  CALL HG(S(1),S(2),C2)
  FUNC(1)=XD(HK)-XDHK
  FUNC(2)=XB(LK)-XBLK
  DO 10 J=1,2
  10 DG(J,I)=(FUNC(J)-G(J))/DELTA
  20 S(I)=S(I)-DELTA
C
C** UPDATE DFLK AND BFHK
C
  DM=DG(1,1)*DG(2,2)-DG(1,2)*DG(2,1)
  DELS1=(G(2)*DG(1,2)-G(1)*DG(2,2))/DM
  DELS2=(G(1)*DG(2,1)-G(2)*DG(1,1))/DM
  S(1)=S(1)+DELS1
  S(2)=S(2)+DELS2
  IF(S(1).GT.1.) S(1)=.99999
  IF(S(2).GT.1.) S(2)=.99999
  100 CONTINUE
  WRITE(6,914)
  GO TO 500
C
C** PRINT DISTRIBUTION OF COMPONENTS.
C
  200 WRITE(6,908)
  WRITE(6,909)K,S(1),S(2)
  WRITE(6,910)(I,D(I),XD(I),B(I),XB(I), I=1,NOC)
  WRITE(6,955)SUMD,SUMB
C
C** PRINT NMIN BY FENSKE'S EQUATION. NMIN IS EQUIVALENT
C** TO C2 IN GEDDES EQUATION.
C
  WRITE(6,911)NMIN
C** CALCULATE RMIN BY THE UNDERWOOD'S EQUATIONS.
C
  CALL UWD(Q,PHI,RMIN)
  WRITE(6,912)PHI,RMIN
C
C** FIND NO. OF T.P. AT SPECIFIED R BY GILLILAND'S CORRELATION.
C
  R=RRM*RMIN
  CALL GLLD(R,RMIN,N,NMIN)
  WRITE(6,913)R,N
C
C** DETERMINE FEED PLATE LOCATION BY KIRKBRIDE'S EQUATION
C
  CALL KIRKB(N,NF)
  WRITE(6,915)NF
C
C
C  OUTPUT FORMAT
C
  900 FORMAT(/10X,'*** INPUT ***')
  901 FORMAT(/5X,'TOTAL NO. OF COMPONENTS =',I5)

```

```

902 FORMAT(5X,'LIGHT KEY IS COMPONENT NO.',I5/
+ 5X,'HEAVY KEY IS COMPONENT NO.',I5)
903 FORMAT(5X,'FEED THERMAL CONDITION Q=',F7.1)
920 FORMAT(5X,'THE OPERATING REFLUX RATIO IS',F6.2,
+ 'TIMES RMIN')
904 FORMAT(/5X,'THE FEED COMPOSITION AND RELATIVE '
+ 'VOLATILITIES ARE' //4X,'I',5X,'XF(I)',3X,'ALFA(I)')
905 FORMAT(I5,2F10.4)
906 FORMAT(/5X,'THE SPECIFIED MOLE FRACTIONS OF THE '
+ 'KEYS IN THE DISTILLATE'/5X, 'AND IN THE BOTTOMS ARE'//
+ 10X,'XD(HK)=' ,F10.4,5X,'XB(LK)=' ,F10.4)
907 FORMAT(/5X,' THE INITIALLY GUESSED VALUES OF THE'
+ ' DISTRIBUTIONS ARE'//10X,'DFLK =' ,F10.4,5X,
+ 'BFHK =' ,F10.4)
908 FORMAT(/10X,'*** OUTPUT ****')
909 FORMAT(/5X,'AFTER',I5,' ITERATIONS THE ACCEPTED',
+ ' VALUES OF THE DISTRIBUTIONS ARE'//
+ 10X,'DFLK =' ,F10.4,5X,'BFHK =' ,F10.4)
910 FORMAT(4X,'I',8X,'D(I)',11X,'XD(I)',10X,'B(I)',
+ 11X,'XB(I)')/(I5,4E15.5))
911 FORMAT(/5X,'MINIMUM NUMBER OF PLATES =' ,F7.2)
912 FORMAT(/5X,'PHI =' ,F10.6,5X,'RMIN =' ,F7.3)
913 FORMAT(/5X, 'AT R =' ,F7.3,5X,'NO. OF T.P. =' ,F7.3)
914 FORMAT(/5X,'KMAX EXCEEDED')
915 FORMAT(/5X,'FEED PLATE LOCATION =' ,F7.2)
955 FORMAT(2X,'TOTAL',E13.5,15X,E15.5)
500 STOP
END
C
C
C
SUBROUTINE HG(DFLK,BFHK,C2)
INTEGER LK,HK
COMMON/BLOCK1/XF(15),XD(15),ALFA(15)
COMMON/BLOCK2/FD,SUMB,SUMD,D(15),B(15),F(15),XB(15)
COMMON NOC,LK,HK
C
C** ESTIMATE DISTRIBUTION OF COMPONENTS IN THE DISTILLATE
C** AND THE BOTTOMS BY HENGSTEBECK-GEDDES METHOD.
C
DO 10 I=1,NOC
10 F(I)=FD*XF(I)
D(LK)=DFLK*F(LK)
B(LK)=F(LK)-D(LK)
D(HK)=(1.-BFHK)*F(HK)
B(HK)=F(HK)-D(HK)
C1=ALOG(D(HK)/B(HK))
C2=ALOG(D(LK)/D(HK)*B(HK)/B(LK))/ALOG(ALFA(LK))
DO 20 I=1,NOC
IF (I.EQ.LK) GO TO 20
IF (I.EQ.HK) GO TO 20
B(I)=F(I)/(1.+EXP(C1+C2*ALOG(ALFA(I))))
D(I)=F(I)-B(I)
20 CONTINUE
C
C** CALCULATE THE MOLE FRACTIONS.
C
SUMD=0.
SUMB=0.
DO 30 I=1,NOC
SUMD=SUMD+D(I)
30 SUMB=SUMB+B(I)
DO 40 I=1,NOC
XD(I)=D(I)/SUMD
40 XB(I)=B(I)/SUMB
RETURN

```

```

END
C
C
C
SUBROUTINE UWD(Q,PHI,RMIN)
C
C** THIS SUBROUTINE CALCULATES RMIN BY THE UNDERWOOD'S
C** EQUATIONS. THE VALUE OF PHI LYING BETWEEN THE ALFA'S
C** OF THE KEYS IS OF OBTAINED BY THE BISECTION METHOD
C** PHILL = LOWER LIMIT OF PHI
C** PHIUL = UPPER LIMIT OF PHI
C
INTEGER LK,HK
COMMON/BLOCK1/XF(15),XD(15),ALFA(15)
COMMON NOC,LK,HK
PHILL=ALFA(HK)
PHIUL=ALFA(HK-1)
DO 10 I=1,20
PHI=0.5*(PHILL+PHIUL)
FPHI=Q-1.
DO 20 J=1,NOC
20 FPHI=FPHI+ALFA(J)*XF(J)/(ALFA(J)-PHI)
IF(FPHI) 30,50,40
30 PHILL=PHI
GO TO 10
40 PHIUL=PHI
10 CONTINUE
50 RMIN=-1.
DO 60 I=1,NOC
60 RMIN=RMIN+ALFA(I)*XD(I)/(ALFA(I)-PHI)
RETURN
END
C
C
C
SUBROUTINE GLLD(R,RMIN,N,NMIN)
REAL N,NMIN
X=(R-RMIN)/(R+1.)
Y=1.-EXP((1.+54.4*X)*(X-1.)/((11.+117.2*X)*SQRT(X)))
N=(Y+NMIN)/(1.-Y)
RETURN
END
C
C
C
SUBROUTINE KIRKB(N,NF)
REAL N,NF
INTEGER LK,HK
COMMON/BLOCK1/XF(15),XD(15),ALFA(15)
COMMON/BLOCK2/FD,SUMB,SUMD,D(15),B(15),F(15),XB(15)
COMMON NOC,LK,HK
FEED=(SUMB/SUMD*XF(HK)/XF(LK)*(XB(LK)/XD(HK))**2.)**0.206
NF=FEED*N/(1.+FEED)
RETURN
END

```